Графы

Пусть на плоскости даны точки – назовем их городами, они соединены между собой дорогами. Такие рисунки известны под общим названием графы. Города назовем вершинами графов, дороги ребрами. Такое представление графа слишком уж примитивен, но такое представление графа дает его наглядное представление на плоскости. 

Определим граф как конечное множество вершин V и набор E неупорядоченных и упорядоченных пар вершин и обозначим G=(V,E). Неупорядоченная пара вершин называется ребром, а упорядоченная пара - дугой. Граф, содержащий только ребра, называется неориентированным; граф, содержащий только дуги, - ориентированным, или орграфом. Вершины, соединенные ребром, называются смежными. Ребра, имеющие общую вершину, также называются смежными. Ребро и любая из его двух вершин называются инцидентными. Говорят, что ребро (u, v) соединяет вершины u и v. Каждый граф можно представить на плоскости множеством точек, соответствующих вершинам, которые соединены линиями, соответствующими ребрам. В трехмерном пространстве любой граф можно представить таким образом, что линии (ребра) не будут пересекаться. 

Способы  описания. Выбор соответствующей структуры данных для представления графа имеет принципиальное значение при разработке эффективных алгоритмов. При решении задач используются следующие четыре основных способа описания графа: матрица инциденций;  матрица смежности;  списки связи и перечни ребер. Мы будем использовать только два: матрицу смежности и перечень ребер.


Матрица смежности  - это двумерный массив размерности N*N.

A[i,j]=


Для хранения перечня ребер необходим двумерный массив R  размерности M*2. Строка массива описывает ребро.

Пусть дан граф. Представим данный граф в языке Паскаль в виде матрицы смежности и перечня ребер. 

uses crt;

const n=6;

a:array[1..n,1..n] of integer=

((0,2,0,0,3,0),

 (0,0,0,0,0,3),

 (0,0,0,0,0,2),

 (5,0,0,0,0,4),

 (0,6,0,0,0,0),

 (0,0,0,0,0,0));

var matpreber:array[1..3,1..20] of integer;

i,k,j:integer;

procedure matreb;

begin

  k:=0;

  for i:=1 to n do

    for j:=1 to n do

       if a[i,j]<>0 then begin

                           k:=k+1;

                           matpreber[1,k]:=i;

                           matpreber[2,k]:=j;

                           matpreber[3,k]:=a[i,j];

                         end;

end;

procedure print;

begin

    for i:=1 to 3 do

      begin

        for j:=1 to k do

           write(matpreber[i,j],' ');

        writeln;

      end;

end;

begin

  clrscr;

   matreb;

   print;

  readkey;

end.

Массив а является для данного графа массивом смежности с указанием веса ребер. Массив matpreber является массивом перечня ребер, который получен из матрицы смежности. K есть степенью графа, т.е число инцидентных ребер.  Этими двумя двумерными массивами можно идентифицировать любой граф. 

Перед началом  рассмотрения основных алгоритмов обработки графов введем основные определения в графах. Если в графе имеется ребро (u,v), говорят, что вершина v смежна с вершиной u. Для неориентированного графа отношение смежности является симметричным, но для ориентированных графов это необязательно. Степенью вершины в неориентированном графе называется число инцидентных ей ребер.  Для ориентированного графа различают исходящую степень, определяемую как  число выходящих из него ребер, и входящую степень, определяемую как число входящих в нее ребер. Сумма исходящей и входящей степеней называется степенью вершины. Путь называется простым, если все вершины в нем различны. Подпуть  - это некоторое количество идущих подряд вершин этого пути. Циклом в ориентированном графе называется путь, в котором начальная вершина совпадает с конечным и который содержит хотя бы одно ребро. Цикл называется простым, если в нем нет одинаковых вершин (кроме первой и последней),т.е. если все вершины различны. Граф, в котором нет циклов называется ациклическим.  Неориентированный граф называется связным, если для любой пары вершин существует путь из одной в другую. Ориентированный граф называется сильно связанным, если из любой вершины достижима любая другая.

Поиск в глубину

Идея метода. Поиск начинается с некоторой фиксированной вершины v. Рассматривается  вершина u, смежная с v. Она выбирается. Процесс повторяется с вершиной u. Если на очередном шаге мы работаем с вершиной q и нет вершин, смежных с q и не рассмотренных ранее (новых), то возвращаемся из вершины q к вершине, которая была до нее. В том случае, когда это вершина v, процесс просмотра закончен. Очевидно, что для фиксации признака, просмотрена вершина графа или нет, требуется структура данных типа:

Nnew : array[1..N] of boolean.  

Пример. 
Пусть граф описан матрицей смежности A. Поиск начинается с первой вершины. На левом рисунке приведен исходный граф, а на правом рисунке у вершин в скобках указана та очередность, в которой вершины графа просматривались в процессе поиска в глубину. 


Логика.
procedure Pg(v:integer);{Массивы Nnew и A глобальные}


 var j:integer;

 begin


Nnew[v]:=false; write(v:3);


for j:=1 to N do if (A[v,j]<>0) and Nnew[j] then Pg(j);

 end;

Используя метод поиска путей  в графе  в глубину  можно написать программу, которая ищет всевозможные пути в графе от заданной вершины к другой. В представленной ниже программе используется «перебор всех возможных вариантов с возвратом».

Рассмотрим программу.

uses crt;

var a:array[1..200,1..200] of byte;

    bil:array[1..200] of boolean;

    put:array[1..300] of integer;

    i,n,j,p,na,kon,dl,odl,kz:integer;

    kol:integer;

    f:text;

procedure enter;

begin

   assign(f,'input.txt');

   reset(f);

   read(f,n);

   for i:=1 to n do

     begin

       for j:=1 to n do

           read(f,a[i,j]);

     end;

   close(f);

end;

procedure print;

begin

   for i:=1 to n do

     begin

       for j:=1 to n do

          write(a[i,j],' ');

       writeln;

     end;

end;

procedure poisk(ng,kg,c:integer);

var ii:integer;

begin

   if ng=kg then begin  {если дошли до конца пути, то выводим путь}

                   write('Путь:');

                   for ii:=1 to c-1 do

                       write(put[ii],' ');

                   writeln;

                  end

   else begin { ищем промежуточные путь после ng}

          for ii:=1 to n do

           if (a[ng,ii]<>0) and (bil[ii]<>true){вершины связаны и текущая вершина }

           then begin           {не рассмотрена}

                   put[c]:=ii;

                   bil[ii]:=true;

                   poisk(ii,kg,c+1);{ищем путь до конечной вершины (рекурсивно)}

                   put[c]:=0;      {c  вершины ii могут быть другие пути,}

                   bil[ii]:=false;{поэтому открываем путь от вершины ii}

                end;

        end;

end;

procedure enterparam;

begin

   write('введите начальную и конечную вершины');

   readln(na,kon);

   fillchar(bil,sizeof(bil),false);

   put[1]:=na;    {запомним начало пути}

   bil[na]:=true; {вершина рассмотрена}

   p:=1;          {1 - ое звено найдено }

   poisk(na,kon,p+1); {ищем следующее (2-ое звено)}

end;

begin

   clrscr;

   enter;

   print;

   enterparam;

   readkey;

end.

В этой программе граф вводится из файла input.txt. В матрице смежности а

A[i,j]=


Массив bil    содержит в начале значение  логическое  false,  означает, что еще ни одна вершина не просмотрена. Массив put содержит вершины, через которые нужно пройти от na до kon, где na- начало вершину от которой ищется путь, а kon- конец вершины до которой ищется путь.

Выбор всех путей производит процедура poisk(ng,kg,c), которая вызывается рекурсивно. Путь ищется от вершины ng до вершины kg, при этом путь запоминается в массиве put. Переменная с является переменной, которая запоминает номер вершины, которая найдена в промежутке от  ng  до kg. Первая вершина пути совпадает с началом графа. Далее ищется следующее звено   пути в графе. Если путь от начальной вершины к конечной есть ищем следующий путь, до тех пор пока не наткнемся на вершину где нет дальше путей. 

Используя поиск в глубину можно найти длину минимального пути в графе между заданными вершинами. Программа решающая данную задачу представлена ниже.

uses crt;

var a:array[1..200,1..200] of byte;

    bil:array[1..200] of boolean;

    put,d:array[1..300] of integer;

    i,n,j,p,na,kon,dl,odl,kz:integer;

    f:text;

procedure enter;

begin

  assign(f,'input.txt');

  reset(f);

  read(f,n);

  for i:=1 to n do

    begin

      for j:=1 to n do

        read(f,a[i,j]);

    end;

  close(f);

end;

procedure print;

begin

  for i:=1 to n do

    begin

      for j:=1 to n do

        write(a[i,j],' ');

        writeln;

    end;

end;

procedure poisk(ng,kg,c:integer);

var ii:integer;

begin

  if ng=kg then begin

                  write('Путь:');  dl:=odl;

                    for ii:=1 to c-1 do

                       begin

                         d[ii]:=put[ii];

                         kz:=c-1;

                         write(put[ii],' ');

                       end;

                    writeln('dl=',dl)

                 end

  else begin

         for ii:=1 to n do

           if (a[ng,ii]<>0) and (bil[ii]<>true)

              and((dl=0)or(odl+a[ng,ii]<dl))

           then begin

                  put[c]:=ii;

                  bil[ii]:=true;

                  odl:=odl+a[ng,ii];

                  poisk(ii,kg,c+1);

                  put[c]:=0;

                  bil[ii]:=false;

                  odl:=odl-a[ng,ii];

                end;

       end;

end;

procedure enterparam;

begin

   write('введите начальную и конечную вершины');

   readln(na,kon);

   fillchar(bil,sizeof(bil),false);

   put[1]:=na;

   bil[na]:=true;

   p:=1; dl:=0;

   poisk(na,kon,p+1);

end;

begin

   clrscr;

   enter;

   print;

   enterparam;

   write('Путь:');

   for i:=1 to kz do

     write(d[i],' ');

   writeln('длина равна ',dl);

   readkey;

end.

 Идея алгоритма в точности совпадает с поиском всех путей в графе. Дополнительно вводится массив, где запоминается минимальный путь. Это массив d. Для рекурсивного поиска минимального пути вводится дополнительное условие, накапливаемый путь должен  быть минимальным.

Поиск в ширину

Идея метода. Суть (в сжатой формулировке) заключается в том, чтобы рассмотреть все вершины, связанные с текущей. Принцип выбора следующей вершины - выбирается та, которая была раньше рассмотрена. Для реализации данного принципа необходима структура данных “очередь”. 


Пример. Исходный граф на левом рисунке. На правом рисунке рядом с вершинами в скобках указана очередность просмотра вершин графа.


Приведем процедуру реализации данного метода обхода вершин графа.


Логика просмотра вершин.

procedure PW(v:integer);

  var 
Og:array[1..N] of 0..N;
                 {очередь}


yk1,yk2:integer;
                             {указатели очереди, yk1 - запись; yk2 - чтение}


j:integer;


   begin

      FillChar(Og,SizeOf(Og),0);yk1:=0;yk2:=0; {начальная инициализация}

      Inc(yk1);Og[yk1]:=v;Nnew[v]:=false;         {в очередь - вершину v}

      while yk2<yk1 do begin
                             {пока очередь не пуста}

      Inc(yk2);v:=Og[yk2];write(v:3);                  {“берем” элемент из очереди}

       for j:=1 to N do                                            {просмотр всех вершин, связанных с вершиной v}

          if (A[v,j]<>0) and Nnew[j] then begin     {если вершина ранее не просмотрена}


 Inc(yk1);Og[yk1]:=j;Nnew[j]:=false;    {заносим ее в очередь}



end;


end;

   end;


Деревья.


Деревом называют произвольный связный неориентированный граф без циклов.  Его можно определить и по-другому: связный граф, содержащий N вершин и N-1 ребер, либо граф, в котором каждая пара вершин соединена одной и только одной простой цепью. Для произвольного связного неориентированного графа G=<V,E> каждое дерево <V,T>, где T(E называют стягивающим деревом (каркасом, остовом). Ребра такого дерева называют ветвями, а остальные ребра графа - хордами. 


 
Число различных каркасов полного связного неориентированного помеченного графа с N вершинами было найдено впервые Кэли и равно N(N-2).


Пример. Граф и его каркасы.

Поиск стягивающего дерева (каркаса). Рассмотрим два алгоритма нахождения каркасов (одного), основанных на методах просмотра графа поиском в глубину и в ширину.  Граф описывается матрицей смежности A, массив Nnew (array[1..N] of boolean) служит для хранения признаков. Значение Nnew[i], равное true, говорит о том, что вершина с номером i еще не просмотрена.  Для хранения ребер, образующих каркас, будем использовать структуру данных Tree (array[1..2,1..N] of integer). 


Известно, что как при поиске в глубину, так и при поиске в ширину просматриваются все вершины связного графа. Использование массива Nnew обеспечивает “подключение” очередного ребра к каркасу без образования циклов. Действительно, цикл образуется, если мы соединяем две просмотренные вершины. Но в нашем случае “подключается” ребро, соединяющее просмотренную вершину с не просмотренной. И наконец, по самой логике методов поиска в глубину и в ширину мы строим связный граф. 


Построение каркаса для связного графа методом поиска в глубину.

procedure Tree_Depth(v:integer); {рекурсивная процедура}


{A, Nnew, Tree, yk - глобальные структуры данных}

  var i:integer;


begin


  Nnew[v]:=false;


  for i:=1 to N do if (A[v,i]<>0) and Nnew[i] then begin





{добавляем ветвь в каркас}





 Inc(yk);Tree[1,yk]:=v;Tree[2,yk]:=i;





 Tree_Depth(i);





end;


end;

{фрагмент основной логики}

begin


FillChar(Nnew,SizeOf(Nnew),true);yk:=0;


Tree_Depth(1); {строим от первой вершины}


<вывод каркаса>;

end.


Построение каркаса для связного графа методом поиска в ширину.

procedure Tree_Width(v:integer);

{A, Tree, yk - глобальные структуры данных}

  var  Nnew:array[1..N] of boolean;


Turn:array[1..N] of integer;yr,yw,i:integer;

 begin

   FillChar(Nnew,SizeOf(Nnew),true);

   FillChar(Turn,SizeOf(Turn),0);yr:=0;yw:=0;

   Inc(yw); Turn[yw]:=v;Nnew[v]:=false;

   while yw<>yr do begin


Inc(yr);v:=Turn[yr];


for i:=1 to N do  if (A[v,i]<>0) and Nnew[i] then begin





 Inc(yw);Turn[yw]:=i;Nnew[i]:=false;





 Inc(yk);Tree[1,yk]:=v;Tree[2,yk]:=i;





end;

    end;

 end;


Пример. Граф и его каркасы, построенные методами поиска в глубину и в ширину. В круглых скобках указана очередность просмотра вершин графа при соответствующем поиске.

Циклы

3.5.1. Эйлеровы циклы


Определение.  Эйлеров цикл — это такой цикл, который проходит ровно один раз по каждому ребру.


Теорема. Связный неориентированный граф G содержит эйлеров цикл тогда и только тогда, когда число вершин нечетной степени равно нулю.


Не все графы имеют эйлеровы циклы, но если эйлеров цикл существует, то это означает, что , следуя вдоль этого цикла, можно нарисовать граф на бумаге, не отрывая от нее карандаша.  Дан граф G, удовлетворяющий условию теоремы. Требуется найти эйлеров цикл. Используется просмотр графа методом поиска в глубину, при этом ребра удаляются. Порядок просмотра (номера вершин)  запоминается. При обнаружении вершины, из которой не выходят ребра, мы их удалили,  ее номер записывается в стек, и просмотр продолжается от предыдущей вершины. Обнаружение вершин с нулевым числом ребер говорит о том, что найден цикл. Его можно удалить, четность вершин (количество выходящих ребер ) при этом не изменится. Процесс продолжается до тех пор, пока есть ребра. В стеке после этого будут записаны номера вершин графа в порядке, соответствующем Эйлерову циклу.


Логика.
procedure Search(v:integer);

{*глобальные переменные*}

{*A - матрица смежности, Сv - стек*}

{* yk - указатель стека*}

 var j:integer;

 begin 



for j:=1 to N do if A[v,j]<>0 then begin





A[v,j]:=0;A[j,v]:=0;Search(j)





end;


Inc(yk);


Cv[yk]:=v;

 end;


Пример графа и содержимое стека Cv после работы процедуры Search.

3.5.2. Гамильтоновы циклы

Определение. Граф называется гамильтоновым, если в нем имеется цикл, содержащий каждую вершину этого графа. Сам цикл также называется гамильтоновым.


Не все связные графы гамильтоновы.


Дан связный неориентированный граф G. Требуется найти все гамильтоновы циклы графа, если они есть.


Метод решения - перебор с возвратом (backtracking). Начинаем поиск решения, например, с первой вершины графа. Предположим, что уже найдены первые k компонент решения. Рассматриваем ребра, выходящие из последней вершины. Если есть такие, что идут в ранее не просмотренные вершины, то включаем эту вершину в решение и помечаем ее как просмотренную. Получена (k+1) компонента решения. Если такой вершины нет, то возвращаемся к предыдущей вершине и пытаемся найти ребро из нее, выходящее в другую вершину. Решение получено при просмотре всех вершин графа и возможности достичь из последней первой вершины. Решение (цикл) выводится, и продолжается процесс нахождения следующих циклов. Пример графа и часть дерева, показывающего механизм работы данного метода.

Логика решения.

procedure Gm(k:integer); {* k - номер итерации*}

                                          {*глобальные переменные*}

{* A - матрица смежности, St - массив для хранения порядка просмотра вершин графа, 

Nnew - массив признаков: вершина просмотрена или нет*}

  var j,v:integer;

  begin 


v:=St[k-1]; {*номер последней вершины*}


for j:=1 to N do if (A[v,j]<>0)  then {*есть ребро между 








вершинами с номерами v и j*}





if (k=N+1) and (j=1) then <вывод цикла>





  else if Nnew[j] then {*вершина не 









просмотрена*}






begin 

                                                               St[k]:=j;Nnew[j]:=false;






    Gm(k+1);Nnew[j]:=true;

                                                           end;

   end;  


Фрагмент основной логики.

......

St[1]:=1;Nnew[1]:=false; Gm(2);

Кратчайшее расстояние от вершины нач до остальных вершин.

(Алгоритм Дейкстры).

Обозначения:  C[i,j]- длина ребра(i,j),  С[i,j]>=0 (если ребра нет,  то его длина полагается равной бесконечности).      D[i]- кратчайшее текущее расстояние от вершины нач до вершины i.      flag[i]- информация о просмотре вершины i:  0 - если  вершина не просмотрена, 1 - если просмотрена. Если вершина просмотрена, то для нее  D[i] есть наикратчайшее расстояние от вершины нач до вершины i.     p[i]- информация о номере вершины, предшествующей вершине i в кратчайшем пути от вершины нач.      minras - это минимальное расстояние.

uses crt;

var c:array[1..50,1..50] of longint;

    flag:array[1..50] of byte;

    d,p:array[1..50] of longint;

    f:text;

    i,j,n,NA,k:longint;

    minras:longint;

procedure enter;

begin

  assign(f,'input.txt');

  reset(f);

  read(f,n);

  for i:=1 to n do

    for j:=1 to n do

      read(f,c[i,j]);

  close(f);

end;

procedure preobr;

begin

  for i:=1 to n do

    for j:=1 to n do

      if c[i,j]=0 then c[i,j]:=maxint;

  for i:=1 to n do

   c[i,i]:=0;

end;

procedure print;

begin

  writeln;

  for i:=1 to n do

    begin

      for j:=1 to n do

        begin

          if c[i,j]=maxint then write(0:7) else write(c[i,j]:7);

        end;

       writeln;

    end;

end;

procedure work;

begin

  write('введите НАЧАЛО ГРАФА');

  READLN(NA);

  FOR I:=1 TO N DO

    BEGIN
      p[i]:=na;       {знаем только начало миимального пути }

      flag[i]:=0;     {ещё ни одна вершина не рассмотрена   }

      d[i]:=c[na,i];  {запомним все расстояния от начала на }

    END;

  flag[na]:=1;       { пока мы знаем только расстояние     }

  p[na]:=0;          { от вершины na до нее же, равное 0   }

  for i:=1 to n-1 do

    begin

      minras:=maxint;

      for j:=1 to n do

        begin

          if (flag[j]=0)and(minras>d[j]) { находим минимальное    }

           then begin                    {расстояние              }

                  minras:=d[j];k:=j;     { до непомеченных вершин }

                  end;

           end;

       flag[k]:=1;            { вершина k помечается просмотренной }

       for j:=1 to n do       { выполняем просмотр                 }

          begin

            if (flag[j]=0) and(d[j]>d[k]+c[k,j])

            then

              begin               {Т.е. если для вершины j еще не найдено  }

                flag[k]:=1;       {кратчайшее расстояние от na и из вершины}

                d[j]:=d[k]+c[k,j] {k по дуге c[k,j] путь в j короче, чем   }

              end;                {чем найденный ранее,то запоминаем его   }

          end;

    end;

end;

procedure printput;

begin

   writeln('Минимальное расстояние от ',na,' до остальных вершин=');

   for i:=1 to n do

     write(d[i],' ');

end;

begin

  clrscr;

  enter;

  preobr;

  print;

  work;

  printput;

  readkey;

end.

 Кратчайшие пути между всеми парами вершин. 

   Если Алгоритм Дейкстры применить к каждой вершине, то получаем кратчайшие пути между всеми парами. Программа при этом будет имеет вид:

uses crt;

var c:array[1..50,1..50] of longint;

    flag:array[1..50] of byte;

    d,p:array[1..50] of longint;

    f:text;

    i,j,n,NA,k,kk:longint;

    minras:longint;

procedure enter;

begin

  assign(f,'input.txt');

  reset(f);

  read(f,n);

  for i:=1 to n do

    for j:=1 to n do

      read(f,c[i,j]);

  close(f);

end;

procedure preobr;

begin

  for i:=1 to n do

    for j:=1 to n do

      if c[i,j]=0 then c[i,j]:=maxint;

  for i:=1 to n do

   c[i,i]:=0;

end;

procedure print;

begin

  writeln;

  for i:=1 to n do

    begin

      for j:=1 to n do

        begin

          if c[i,j]=maxint then write(0:7) else write(c[i,j]:7);

        end;

       writeln;

    end;

end;

procedure work;

begin

  for kk:=1 to n do

  begin

  FOR I:=1 TO N DO

    BEGIN

      p[i]:=kk;

      flag[i]:=0;

      d[i]:=c[kk,i];

    END;

  flag[kk]:=1;

  p[na]:=0;

  for i:=1 to n-1 do

    begin

      minras:=maxint;

      for j:=1 to n do

        begin

          if (flag[j]=0)and(minras>d[j])

          then begin

                  minras:=d[j];k:=j;

                  end;

           end;

       flag[k]:=1;

       for j:=1 to n do

          begin

            if (flag[j]=0) and(d[j]>d[k]+c[k,j])

            then

              begin

                flag[k]:=1;

                d[j]:=d[k]+c[k,j]

              end;

          end;

    end;

    writeln('Минимальные расстояние от ',kk,' до остальных вершин=');

    for i:=1 to n do

      write(d[i],' ');

      writeln;

   end;

end;

begin

  clrscr;

  enter;

  preobr;

  print;

  work;

  readkey;

end.

Алгоритм Флойда


Дано. Ориентированный граф G=<V,E> с матрицей весов A(array[1..N,1..N] of integer). Результат. Матрица D кратчайших расстояний между всеми парами вершин графа и кратчайшие пути.


Идея алгоритма. Обозначим через Dm[i,j]  оценку кратчайшего пути из i в j с промежуточными вершинами из множества [1..m]. Тогда имеем: D0[i,j]:=A[i,j] и D(m+1)[i,j]=min{Dm[i,j],Dm[i,m+1]+Dm[m+1,j]}.  Второе равенство требует пояснения. Пусть мы находим кратчайший путь из i в j c промежуточными вершинами из множества [1..(m+1)]. Если этот путь не содержит вершину (m+1), то D(m+1)[i,j]=Dm[i,j]. Если же он содержит эту вершину, то его можно разделить на две части от i до (m+1) и от (m+1)  до j. 


Время работы алгоритма пропорционально N3.

{$A+,B-,D+,E+,F-,G-,I+,L+,N+,O-,P-,Q-,R-,S+,T-,V+,X+}

{$M 64384,0,655360}

uses crt;

type mas=array[1..60,1..60] of longint;

var a,d,p:mas;

f:text;

i,j,n,m,c,op,k,max,min,ce,ll,mm,kk,IT:longint;

pt,mput:array[1..50] of longint;

procedure enter;

begin

  assign(f,'input.txt');

  reset(f);

  read(f,n);

  for i:=1 to n do

    for j:=1 to n do

      read(f,a[i,j]);

   close(f);

  for i:=1 to n do

    for j:=1 to n do

       begin

          d[i,j]:=a[i,j];

          if a[i,j]=0 then d[i,j]:=maxint;

          p[i,j]:=i;

       end;

end;

procedure print(m:mas);

begin

  for i:=1 to n do

    begin

      for j:=1 to n do

        write(m[i,j]:3);

      writeln;

    end;

   writeln;

end;

procedure floid;

begin

  for i:=1 to n do d[i,i]:=0;

  for m:=1 to n do

    for i:=1 to n do

      for j:=1 to n do

         BEGIN

           if d[i,j]>d[i,m]+d[m,j] then  begin

                                           d[i,j]:=d[i,m]+d[m,j];

                                           p[i,j]:=p[m,j];

                                         end;

                                            IT:=IT+1;

         END;

end;

procedure put(i,j:integer);

begin

  if p[i,j]=i then if i=j then write(i)

                          else begin

                                 inc(c);

                                 pt[c]:=i;

                                 pt[c+1]:=j;

                                end

              else begin

                     put(i,p[i,j]);

                     put(p[i,j],j);

                   end;

end;

procedure wseputi;

begin

  for i:=1 to n do

   for j:=1 to n do

     begin

       if i<>j then begin

                      c:=0;

                      textcolor(5);

                      fillchar(pt,sizeof(pt),0);

                      put(i,j);inc(c);pt[c]:=j;

                      write('от ',i,' до ',j,' ');

                      textcolor(6);

                      for m:=1 to c do

                        write(pt[m],' ');

                      op:=0;

                      for k:=1 to c-1 do

                        op:=op+d[pt[k],pt[k+1]];

                      textcolor(3);

                      write('длина пути=',op,' ');

                    end;

     end;

end;

begin

  clrscr;

  enter;

  floid;

  print(a);

  print(d);

  print(p);

  wseputi;

  readkey;

end.

Примечание. Верхний индекс у D указывает номер итерации (значение m в процедуре floid).

Расстояния между парами вершин дает D. Для вывода самих кратчайших путей введем матрицу p того же типа, что и D. Элемент p[i,j] определяет предпоследнюю вершину кратчайшего пути из i в j.


Процедура floid претерпит небольшие изменения. В том случае, когда D[i,j] больше D[i,m]+D[m,j], изменяется не только D[i,j], но и p[i,j].  p[i,j]  присваивается значение M[m,j]. Для нашего примера изменения M выглядят следующим образом.

Например, необходимо вывести кратчайший путь из 3-й вершины во 2-ю.  Элемент M[3,2] равен 1, поэтому смотрим на элемент M[3,1]. Он равен четырем. Сравниваем M[3,4] c 3-й. Есть совпадение, мы получили кратчайший путь: 3(4(1( 2.

Для вывода пути будем использовать рекурсивную процедуру put.

procedure put(i,j:integer);

begin

  if p[i,j]=i then if i=j then write(i)

                          else begin

                                 inc(c);

                                 pt[c]:=i;

                                 pt[c+1]:=j;

                                end

              else begin

                     put(i,p[i,j]);

                     put(p[i,j],j);

                   end;

end;

Для поиска всех кратчайших путей можно  преобразовать программу, которая при помощи поиска в глубину  находила кратчайший путь между двумя выбранными вершинами.

После некоторых изменений программа прием вид. 

uses crt;

var

a:array[1..100,1..100] of byte;

  put,mind:array[1..100] of byte;

  byl:array[1..100] of boolean;

  i,j,n,na,ko,dlina,k:integer;

  dl,odl:integer;

  f:text;

procedure enter;

begin

   assign(f,'input.txt');

   reset(f);

   read(f,n);

   for i:=1 to n do

     for j:=1 to n do

      read(f,a[i,j]);

   close(f);

end;

procedure print;

begin

   for i:=1 to n do

     begin

        for j:=1 to n do

         write(a[i,j],' ');

        writeln;

     end;

   writeln;

end;

procedure vpvg(ot,ku,z:integer);

var wp,pv:integer;

begin

    INC(IT);

    if ot=ku then begin

                     dl:=odl; dlina:=z-1;

                     for wp:=1 to z-1 do

                       begin

                         mind[wp]:=put[wp];

                       end;

                  end

    else for pv:=1 to n do

           if (a[ot,pv]<>0) and(byl[pv]=false)

           and((dl=0)or((odl+a[ot,pv])<dl))

           then begin

                  put[z]:=pv; byl[pv]:=true;

                  odl:=odl+a[ot,pv];

                  vpvg(pv,ku,z+1);

                  put[z]:=0; byl[pv]:=false;

                  odl:=odl-a[ot,pv];

                end;

end;

begin

   clrscr;

   enter;

   print;

   for i:=1 to n do

      for j:=1 to n do

         if i<>j then begin

                         na:=i;ko:=j;

                         fillchar(put,sizeof(put),0);

                         fillchar(byl,sizeof(byl),false);

                         put[1]:=na;

                         byl[na]:=true;

                         vpvg(na,ko,2);

                         TEXTCOLOR(5);

                         write('Путь от ',i,' до',j,' ');

                         TEXTCOLOR(4);

                          for k:=1 to dlina do

                              write(mind[k], ' ');

                         write('ДЛИНА ПУТИ РАВНА ',DL,' ');

                         DL:=0;

                      END;

   readkey;

end.

Сложность этих алгоритмов примерно одинакова.

